


## ARRAY

Array is a derived or user defined data type, which is used to store more than one values of same data type. It means, it is used to store homogenous types of data.

Hence, in the case, variable name is same for more than one values of same data type.

An array stores a fixed-size sequential collection of elements of the same type. An array is used to store a collection of data, but it is often more useful to think of an array as a collection of variables of the same type.

All arrays consist of contiguous memory locations. The lowest address corresponds to the first element and the highest address to the last element.



### **Syntax: -**

```
Dim a (10) as integer
Dim b (10) as single
Dim c (10) as string
```

### Creating Arrays in VB.Net

To declare an array in VB.Net, you use the Dim statement. For example,

```
Dim intData(30)          ' an array of 31 elements
Dim strData(20) As String ' an array of 21 strings
Dim twoDarray(10, 20) As Integer 'a two dimensional array of integers
Dim ranges(10, 100)      'a two dimensional array
```

You can also initialize the array elements while declaring the array. For example,

```
Dim intData() As Integer = {12, 16, 20, 24, 28, 32}
Dim names() As String = {"Karthik", "Sandhya", _
"Shivangi", "Ashwitha", "Somnath"}
Dim miscData() As Object = {"Hello World", 12d, 16ui, "A"c}
```

There are two types of arrays: -

1. Single Dimensional Array
2. Multi Dimensional Array

## 1. SINGLE DIMENSIONAL ARRAY: -

When one dimension is mentioned with an array variable than it is known as Single Dimensional Array.

Syntax: -

Dim variable (size) As Data type


**Example: -**

Dim a (10) As Integer

The memory is allocated by the array variable name either horizontally or vertically and whole space is divided into different blocks according to the given size. Each block is given the unique numerical number (starts from 0), known as INDEX NUMBER / BLOCK NUMBER / POCKET NUMBER. Each block has the same size and has capacity to hold one number at a time.

**Example: -**

Dim a (5) As Integer



Now, the value is assigned or retrieved to or from an array variable, by using the variable name and index number.

Assignment		Retrieve	
a (2) = 10		print a (2)	
a (4) = 40		print a (4)	

The array variable can also be declared as follows: -

**Syntax: -**

Dim variable (v1 to v2) As Integer

**Example: -**

Dim a (0 to 10) As Integer

0	1	2	3	4	5	6	7	8	9	10

## **2. MULTI- DIMENSIONAL ARRAY: -**

When more than one dimension is mentioned in an array variable, then it is known as Multi – Dimensional array.

They include Double Dimensional, Triple Dimensional etc.

### **Double Dimensional Array: -**

When two dimensions is mentioned with an array variable, and then it is known as Double Dimensional array.

### **Syntax: -**

Dim variable (Size. Size) As Integer


The first —size stands for row value and the second size stands for column value.

The memory space is allocated by the DDA variable name and whole space is divided into different row and column according to specified dimensions. Each row and column is numbered by an index number, starts from Zero (0).

**Example: -**

Dim a (2, 2) As Integer


The memory space is allocated as follows:



The combination of row and column is known as cell. So, the data is stored in DDA according to cell wise.

We can also declare the DDA as follows:

Dim a (0 to 2, 0 to 2) As Integer



The elements in an array can be stored and accessed by using the index of the array. The following program demonstrates this –

```
Module arrayApl
Sub Main()
    Dim n(10) As Integer ' n is an array of 11 integers '
    Dim i, j As Integer
    ' initialize elements of array n '

    For i = 0 To 10
        n(i) = i + 100 ' set element at location i to i + 100
    Next i
    ' output each array element's value '

    For j = 0 To 10
        Console.WriteLine("Element({0}) = {1}", j, n(j))
    Next j
    Console.ReadKey()
End Sub
End Module
```

When the above code is compiled and executed, it produces the following result –

```
Element(0) = 100
Element(1) = 101
Element(2) = 102
Element(3) = 103
Element(4) = 104
Element(5) = 105
Element(6) = 106
Element(7) = 107
Element(8) = 108
Element(9) = 109
Element(10) = 110
```

### Dynamic Arrays

Dynamic arrays are arrays that can be dimensioned and re-dimensioned as per the need of the program. You can declare a dynamic array using the **ReDim** statement.

Syntax for ReDim statement –

ReDim [Preserve] arrayname(subscripts)

Where,

- The **Preserve** keyword helps to preserve the data in an existing array, when you resize it.
- **arrayname** is the name of the array to re-dimension.
- **subscripts** specifies the new dimension.

```

Module arrayApl
Sub Main()
    Dim marks() As Integer
    ReDim marks(2)
    marks(0) = 85
    marks(1) = 75
    marks(2) = 90

    ReDim Preserve marks(10)
    marks(3) = 80
    marks(4) = 76
    marks(5) = 92
    marks(6) = 99
    marks(7) = 79
    marks(8) = 75

    For i = 0 To 10
        Console.WriteLine(i & vbTab & marks(i))
    Next i
    Console.ReadKey()
End Sub
End Module

```

When the above code is compiled and executed, it produces the following result –

0	85
1	75
2	90
3	80
4	76
5	92
6	99
7	79
8	75
9	0
10	0

## Multi-Dimensional Arrays

VB.Net allows multidimensional arrays. Multidimensional arrays are also called rectangular arrays.

You can declare a 2-dimensional array of strings as –

```
Dim twoDStringArray(10, 20) As String
```

or, a 3-dimensional array of Integer variables –

```
Dim threeDIntArray(10, 10, 10) As Integer
```

The following program demonstrates creating and using a 2-dimensional array –

```
Module arrayApl
Sub Main()
    ' an array with 5 rows and 2 columns
    Dim a(,) As Integer = {{0, 0}, {1, 2}, {2, 4}, {3, 6}, {4, 8}}
    Dim i, j As Integer
    ' output each array element's value '

    For i = 0 To 4
        For j = 0 To 1
            Console.WriteLine("a[{0},{1}] = {2}", i, j, a(i, j))
        Next j
    Next i
    Console.ReadKey()
End Sub
End Module
```

When the above code is compiled and executed, it produces the following result –

```
a[0,0]: 0
a[0,1]: 0
a[1,0]: 1
a[1,1]: 2
a[2,0]: 2
a[2,1]: 4
a[3,0]: 3
a[3,1]: 6
a[4,0]: 4
a[4,1]: 8
```

### Jagged Array

A Jagged array is an array of arrays. The following code shows declaring a jagged array named *scores* of Integers –

```
Dim scores As Integer()() = New Integer(5)() { }
```

The following example illustrates using a jagged array –

```
Module arrayApl
Sub Main()
    'a jagged array of 5 array of integers
    Dim a As Integer()() = New Integer(4)() { }
    a(0) = New Integer() {0, 0}
    a(1) = New Integer() {1, 2}
    a(2) = New Integer() {2, 4}
    a(3) = New Integer() {3, 6}
    a(4) = New Integer() {4, 8}
    Dim i, j As Integer
```

```

' output each array element's value

For i = 0 To 4
    For j = 0 To 1
        Console.WriteLine("a[{0},{1}] = {2}", i, j, a(i)(j))
    Next j
Next i
Console.ReadKey()
End Sub
End Module

```

When the above code is compiled and executed, it produces the following result –

```

a[0][0]: 0
a[0][1]: 0
a[1][0]: 1
a[1][1]: 2
a[2][0]: 2
a[2][1]: 4
a[3][0]: 3
a[3][1]: 6
a[4][0]: 4
a[4][1]: 8

```

## The Array Class

The Array class is the base class for all the arrays in VB.Net. It is defined in the System namespace. The Array class provides various properties and methods to work with arrays.

## Properties of the Array Class

The following table provides some of the most commonly used **properties** of the **Array** class –

Sr.No	Property Name & Description
1	<b>IsFixedSize</b> Gets a value indicating whether the Array has a fixed size.
2	<b>IsReadOnly</b> Gets a value indicating whether the Array is read-only.
3	<b>Length</b>



	Gets a 32-bit integer that represents the total number of elements in all the dimensions of the Array.
4	<b>LongLength</b> Gets a 64-bit integer that represents the total number of elements in all the dimensions of the Array.
5	<b>Rank</b> Gets the rank (number of dimensions) of the Array.

### Methods of the Array Class

The following table provides some of the most commonly used **methods** of the **Array** class –

Sr.No	Method Name & Description
1	<b>Public Shared Sub Clear (array As Array, index As Integer, length As Integer)</b> Sets a range of elements in the Array to zero, to false, or to null, depending on the element type.
2	<b>Public Shared Sub Copy (sourceArray As Array, destinationArray As Array, length As Integer)</b> Copies a range of elements from an Array starting at the first element and pastes them into another Array starting at the first element. The length is specified as a 32-bit integer.
3	<b>Public Sub CopyTo (array As Array, index As Integer)</b> Copies all the elements of the current one-dimensional Array to the specified one-dimensional Array starting at the specified destination Array index. The index is specified as a 32-bit integer.
4	<b>Public Function GetLength (dimension As Integer) As Integer</b> Gets a 32-bit integer that represents the number of elements in the specified dimension of the Array.

5	<b>Public Function GetLongLength (dimension As Integer) As Long</b> Gets a 64-bit integer that represents the number of elements in the specified dimension of the Array.
6	<b>Public Function GetLowerBound (dimension As Integer) As Integer</b> Gets the lower bound of the specified dimension in the Array.
7	<b>Public Function GetType As Type</b> Gets the Type of the current instance (Inherited from Object).
8	<b>Public Function GetUpperBound (dimension As Integer) As Integer</b> Gets the upper bound of the specified dimension in the Array.
9	<b>Public Function GetValue (index As Integer) As Object</b> Gets the value at the specified position in the one-dimensional Array. The index is specified as a 32-bit integer.
10	<b>Public Shared Function IndexOf (array As Array,value As Object) As Integer</b> Searches for the specified object and returns the index of the first occurrence within the entire one-dimensional Array.
11	<b>Public Shared Sub Reverse (array As Array)</b> Reverses the sequence of the elements in the entire one-dimensional Array.
12	<b>Public Sub SetValue (value As Object, index As Integer)</b> Sets a value to the element at the specified position in the one-dimensional Array. The index is specified as a 32-bit integer.
13	<b>Public Shared Sub Sort (array As Array)</b> Sorts the elements in an entire one-dimensional Array using the IComparable implementation of each element of the Array.
14	<b>Public Overridable Function ToString As String</b> Returns a string that represents the current object (Inherited from Object).

For complete list of Array class properties and methods, please consult Microsoft documentation.

### Example

The following program demonstrates use of some of the methods of the Array class:

```
Module arrayApl
Sub Main()
    Dim list As Integer() = {34, 72, 13, 44, 25, 30, 10}
    Dim temp As Integer() = list
    Dim i As Integer
    Console.WriteLine("Original Array: ")

    For Each i In list
        Console.WriteLine("{0} ", i)
    Next i
    Console.WriteLine()
    'reverse the array
    Array.Reverse(temp)
    Console.WriteLine("Reversed Array: ")

    For Each i In temp
        Console.WriteLine("{0} ", i)
    Next i
    Console.WriteLine()
    'sort the array
    Array.Sort(list)
    Console.WriteLine("Sorted Array: ")

    For Each i In list
        Console.WriteLine("{0} ", i)
    Next i
    Console.WriteLine()
    Console.ReadKey()
End Sub
End Module
```

When the above code is compiled and executed, it produces the following result –

```
Original Array: 34 72 13 44 25 30 10
Reversed Array: 10 30 25 44 13 72 34
Sorted Array: 10 13 25 30 34 44 72
```